
Auswerten des tatsächlichen 
Reibweges

Problem:
ANSYS bietet die Möglichkeit den Reibweg auszuwerten. Dieser 
Reibweg gibt allerdings immer den direkten Weg zwischen der 
Ausgangslage und der momentanen Position aus. Bewegt sich das 
Bauteil aber nicht auf diesem direkten Weg, kann diese 
Auswertegröße nicht mehr verwendet werden. 

Lösung:
Es wird für jeden einzelnen Schritt, der allerdings in der Ergebnisdatei 
gespeichert werden muß, der inkrementelle Reibweg berechnet und 
aufsummiert. 

Beispiel:
Es wird ein Klotz im ersten Loadstep um 5 in x-Richtung verschoben 
und im zweiten Schritt um 5 in y-Richtung. Die Ausgabe der 
vordefinierten ‚Sliding Distance‘ liefert 7.071.

Berechnung des Reibweges über das Makro:

NAME                              VALUE                        TYPE  DIMENSIONS

SLIDING                           9.99919610                   SCALAR

LT

Eingabe für das Modell 
(ANSYS 9.0):

/prep7
et,1,185
et,2,170
et,3,174
mp,ex,1,210000
mp,prxy,1,0.3
mp,mu,2,0.1
r,2

rect,0,10,0,10
bloc,0,1,0,1,0,1
vmesh,all
type,2
real,2
mat,2
amesh,1
alls,belo
nsel,s,loc,z,0
type,3
esurf,all
/solu
nlgeom,on
nsel,s,loc,z,1
d,all,uz,-0.001
d,all,uy
d,all,ux,5
alls
nsub,8,8,8
outres,all,all
solve
nsel,s,loc,z,1
d,all,ux,5
d,all,uy,5
alls
nsub,10,10,10
outres,all,all
solve
/post1
reibweg

*get,anz,active,,set,nset

esel,s,ename,,174
*get,e_num,elem,,num,min
esel,s,,,e_num

!Berechnen des Reibweges für den ersten Schritt
i=1
set,first

!Auslesen des Reibweges in x-Richtung des Kontaktelements
etab,aasr%i%,nmisc,50

!Auslesen des Reibweges in y-Richtung des Kontaktelements
etab,aass%i%,nmisc,54

!Daraus den resultierenden Reibweg berechnen
smult,srm,aasr%i%,aasr%i%
smult,ssm,aass%i%,aass%i%
sadd,sam,srm,ssm
sexp,sd%i%,sam,,0.5

*get,sliding,elem,e_num,etable,sd1

!Berechnen des Reibweges für alle weiteren Schritte
*do,i,2,anz,1
set,next
etab,aasr%i%,nmisc,50
etab,aass%i%,nmisc,54
sadd,aasr,aasr%i%,aasr%i-1%,,-1
sadd,aass,aass%i%,aass%i-1%,,-1
smult,srm,aasr,aasr
smult,ssm,aass,aass
sadd,sam,srm,ssm
sexp,sd%i%,sam,,0.5
*get,s_step,elem,e_num,etable,sd%i%
sliding=sliding+s_step
*enddo

*stat,sliding

Auswerten des tatsächlichen 
Reibweges

Inhalt des Makros reibweg.mac

Das Makro finden Sie auch unter:
http://www.cadfem.de/fileadmin/files/9_service_newsletter/2005/0506/reibweg.mac

http://www.cadfem.de/fileadmin/files/9_service_newsletter/2005/0506/reibweg.mac

